

## Week 9 – Excel 1

## Q5. Use IF function to check if 2 country regions are equal

```
= IF(C13=C16,"Yes","No")
```

- The subsequent uses of IF() throughout this document are in line with the general definition of IF() in excel: IF(logical\_test,[value\_if\_true],[value\_if\_false])
- It is effectively an if-else block if used as in the complete format above.

## Q6: Is Country #1 and/or Country #2 in the Region specified?

```
= IF(C13=E2,"Yes", IF(C16=E2,"Yes","No"))
```

- If country 1 (C13) is in the region specified, then "Yes", if it is not, we need to check for country 2 (C16), since the question asks if either of the countries are in the specified region.
- We need a second IF(). The result of which would be the [value\_if\_false] in our first IF(). Hence, code to check if country 2 is in the region, is in the ELSE part of the first IF().
- Effectively, you are implementing the following:

```
if (C13==E2)           //country 1 in the region
    return "Yes";
else                   // checking for country 2
{
    if (C16==E2)       // country 2 in the region
        return "Yes";
    else
        return "No";
}
```

## Q7: What was the population in 2000? Country #1:

```
=SUMIFS('Data by country'!J:J , 'Data by country'!A:A , C2 , 'Data by country'!B:B , "7/1/2000")
```

- SUMIFS() general format:  
SUMIFS (sum\_range , criteria\_range1 , criterial ,...)
- For Country #2: Copy and Paste. The cell reference of C2 will change to C3 which is expected.

**Q8: What was the population in 2010?**

```
=SUMIFS('Data by country'!J:J, 'Data by country'!A:A, C2, 'Data by country'!B:B, "7/1/2010")
```

- Copy and paste from Q7.
- When you copy and paste, C2 is changed to C6 which needs to be changed back to C2 and the date to "7/1/2010". Now you can copy paste this formula for Country #2.

**Q9: As a percentage, what was the growth in population from 2000 to 2010?**

```
=(C9-C25)/C25
```

- For percentage change, we use (new value-old value)/old value.
- The data can be formatted by: right clicking on the cell and choosing format cell. We can choose percentage data format here to change the format to percentage.

**Q10: Which country had the larger population growth percentage?**

```
=IF(C33>C34,C2,C3)
```

**Q11: What was the population for each year in 2000 to 2010?**

- **C41:** = C2 and **D41:** = C3
- Copy Pasting the formula is the most efficient option, we would like to write a formula in C42 and then use the same for populating up to C52 and then for D42 to D52.
- ```
=SUMIFS('Data by country'!$J:$J, 'Data by country'!$A:$A, C$41, 'Data by country'!$B:$B, "7/1/" & $B42)
```
- For Joining Strings in Excel: use the '&' sign. We use this in the last argument for SUMIFS () above, to join the day and the year so that when changing rows, the year gets changed automatically using column B.
- Within C41, the rows should be locked as C\$41 so that we always refer to country 1. Which is present in C41. No need to lock the column as that needs to change when we use the same formula for column D, the same applies for column J, A and B (in the last two arguments).

**Q12: What was the year over year growth for each year in 2000 to 2010?**

- **C57:** = (C44-C43)/C43
- Just copy this formula down into all the rows in column C
- Format the cells to percentage number format
- Copy the formula into the column D as well – No locking necessary because the rows and columns will be shifted by 1 which is what we want

## Pattern Matching

- When dealing with strings, you can use pattern matching to match a pattern or substring in a string.
- `*` : To match zero or more characters
- `?` : To match Exactly 1 character
- The string "alice" will match a cell which has exactly alice
- If we need to match alice within alicem, alice abigail, alice2, alice667 etc. we must use "alice\*" as the \* matches with any number of characters.
- "alice\*": will match strings that start with alice
- To match for characters before and after alice, "\*alice\*" : matches all strings that contain alice
- The `?` matches only 1 character – "?un" will match sun, fun, gun, pun
- It will not match trun, brun, wrun etc. Exactly one character is allowed for the '?'
- "?un" will match one character followed by un and nothing else

### i>Clicker Questions:

Pattern: `"?or*"`

Data: world

- Exactly 1 character is matched before "or" and any number of characters are allowed after "or"
- Hence, it matches "world" and "forward" but not "Orange", "preempt" (no "or" present in "preempt") and "realtor" (? Can match with only one character and it goes before "or" in the pattern whereas "realtor" has 5 characters before "or")
- The word "For" will be matched correctly as well, since '\*' is for any number of characters, including 0.
- None of the matching is case sensitive in general.

## Week 9- Excel 2

**Q10. How many days did the comment say only "snow"?**

= `countifs(J4:J33,"snow")` – comment was exactly snow

**Q11. How many days did the comment start with "snow"?**

= `countifs(J4:J33,"snow*")` – comment started with snow

**Q12. How many days did the comment contain "snow"?**

= `countifs(J4:J33,"*snow*")` – comment contains snow